Exercitium Ejercicios de programación funcional con Haskell

Volumen 1: curso 2013-14

José A. Alonso Jiménez

Grupo de Lógica Computacional Dpto. de Ciencias de la Computación e Inteligencia Artificial Universidad de Sevilla

Sevilla, 11 de diciembre de 2018

Esta obra está bajo una licencia Reconocimiento-NoComercial-Compartirlgual 2.5 Spain de Creative Commons.

Se permite:

- copiar, distribuir y comunicar públicamente la obra
- hacer obras derivadas

Bajo las condiciones siguientes:



Reconocimiento. Debe reconocer los créditos de la obra de la manera especificada por el autor.



No comercial. No puede utilizar esta obra para fines comerciales.



Compartir bajo la misma licencia. Si altera o transforma esta obra, o genera una obra derivada, sólo puede distribuir la obra generada bajo una licencia idéntica a ésta.

- Al reutilizar o distribuir la obra, tiene que dejar bien claro los términos de la licencia de esta obra.
- Alguna de estas condiciones puede no aplicarse si se obtiene el permiso del titular de los derechos de autor.

Esto es un resumen del texto legal (la licencia completa). Para ver una copia de esta licencia, visite http://creativecommons.org/licenses/by-nc-sa/2. 5/es/ o envie una carta a Creative Commons, 559 Nathan Abbott Way, Stanford, California 94305, USA.

Para Guiomar

Índice general

1	Iguales al siguiente	9
2	Ordenación por el máximo	13
3	La bandera tricolor	15
4	Elementos minimales	19
5	Mastermind	21
6	Primos consecutivos con media capicúa	25
7	Anagramas	27
8	Primos equidistantes	31
9	Suma si todos los valores son justos	35
10	Matrices de Toeplitz	39
11	Máximos locales	41
12	Lista cuadrada	43
13	Segmentos de elementos consecutivos	47
14	Valor de un polinomio mediante vectores	49
15	Ramas de un árbol	51
16	Alfabeto comenzado en un carácter	55

Introducción

"The chief goal of my work as an educator and author is to help people learn to write beautiful programs."

(Donald Knuth en Computer programming as an art)

Este libro es una recopilación de las soluciones de los ejercicios propuestos en el blog Exercitium ¹ durante el curso 2013–14.

El principal objetivo de Exercitium es servir de complemento a la asignatura de Informática 2 de 1° del Grado en Matemáticas de la Universidad de Sevilla.

Con los problemas de Exercitium, a diferencias de los de las relaciones ³, se pretende practicar con los conocimientos adquiridos durante todo el curso, mientras que con las relaciones están orientadas a los nuevos conocimientos.

Habitualmente de cada ejercicio se muestra distintas soluciones y se compara sus eficiencias.

La dinámica del blog es la siguiente: cada día, de lunes a viernes, se propone un ejercicio para que los alumnos escriban distintas soluciones en los comentarios. Pasado 7 días de la propuesta de cada ejercicio, se cierra los comentarios y se publica una selección de sus soluciones.

Para conocer la cronología de los temas explicados se puede consultar el diario de clase ⁴.

El código del libro se encuentra en GitHub 5

¹https://www.glc.us.es/~jalonso/exercitium

²https://www.cs.us.es/~jalonso/cursos/ilm-13

³https://www.cs.us.es/~jalonso/cursos/ilm-13/ejercicios/ejercicios-I1M-2013.pdf

⁴https://www.glc.us.es/~jalonso/vestigium/category/curso/ilm/ilm2013

⁵https://github.com/jaalonso/Exercitium2013

Iguales al siguiente

Enunciado

Definir la función

```
igualesAlSiguiente :: Eq a => [a] -> [a]
```

tal que (igualesAlSiguiente xs) es la lista de los elementos de xs que son iguales a su siguiente. Por ejemplo,

```
igualesAlSiguiente [1,2,2,2,3,3,4] == [2,2,3]
igualesAlSiguiente [1..10] == []
```

```
import Data.List (group)
import Test.QuickCheck

-- 1@ definición (por comprensión):
igualesAlSiguiente :: Eq a => [a] -> [a]
igualesAlSiguiente xs =
  [x | (x,y) <- zip xs (tail xs), x == y]

-- 2@ definición (por recursión):
igualesAlSiguiente2 :: Eq a => [a] -> [a]
```

```
igualesAlSiguiente2 (x:y:zs)
  | x == y = x : igualesAlSiguiente2 (y:zs)
  | otherwise = igualesAlSiguiente2 (y:zs)
igualesAlSiguiente2 = []
-- 3ª definición (con concat y comprensión):
igualesAlSiguiente3 :: Eq a => [a] -> [a]
igualesAlSiguiente3 xs = concat [ys | (:ys) <- group xs]</pre>
-- 4ª definición (con concat y map):
igualesAlSiguiente4 :: Eq a => [a] -> [a]
igualesAlSiguiente4 xs = concat (map tail (group xs))
-- 5ª definición (con concatMap):
igualesAlSiguiente5 :: Eq a => [a] -> [a]
igualesAlSiguiente5 xs = concatMap tail (group xs)
-- 6ª definición (con concatMap y sin argumentos):
igualesAlSiguiente6 :: Eq a => [a] -> [a]
igualesAlSiguiente6 = concatMap tail . group
-- Equivalencia
- - =========
-- La propiedad es
prop igualesAlSiguiente equiv :: [Int] -> Bool
prop igualesAlSiguiente equiv xs =
  iqualesAlSiguiente xs == iqualesAlSiguiente2 xs &&
  igualesAlSiguiente xs == igualesAlSiguiente3 xs &&
  iqualesAlSiquiente xs == iqualesAlSiquiente4 xs &&
  igualesAlSiguiente xs == igualesAlSiguiente5 xs &&
  iqualesAlSiguiente xs == iqualesAlSiguiente6 xs
verifica igualesAlSiguiente equiv :: 10 ()
verifica igualesAlSiguiente equiv =
  quickCheck prop igualesAlSiguiente equiv
-- La comprobación es
    λ> verifica igualesAlSiguiente equiv
     +++ OK, passed 100 tests.
```

-- (0.07 secs, 9,911,528 bytes)

Ordenación por el máximo

Enunciado

Definir la función

```
ordenadosPorMaximo :: Ord a => [[a]] -> [[a]]
```

tal que (ordenadosPorMaximo xss) es la lista de los elementos de xss ordenada por sus máximos. Por ejemplo,

```
ghci> ordenadosPorMaximo [[3,2],[6,7,5],[1,4]]
[[3,2],[1,4],[6,7,5]]
ghci> ordenadosPorMaximo ["este","es","el","primero"]
["el","primero","es","este"]
```

```
import Data.List (sort)
import GHC.Exts (sortWith)
import Test.QuickCheck

-- 1º definición
ordenadosPorMaximo :: Ord a => [[a]] -> [[a]]
ordenadosPorMaximo xss =
  map snd (sort [(maximum xs,xs) | xs <- xss])</pre>
```

```
-- 2ª definición
ordenadosPorMaximo2 :: Ord a => [[a]] -> [[a]]
ordenadosPorMaximo2 xss =
  [xs | (_,xs) <- sort [(maximum xs,xs) | xs <- xss]]
-- 3ª definición:
ordenadosPorMaximo3 :: Ord a => [[a]] -> [[a]]
ordenadosPorMaximo3 = sortWith maximum
-- Equivalencia
-- =========
verificaOrdenadosPorMaximo :: IO ()
verificaOrdenadosPorMaximo =
 quickCheck prop_ordenadosPorMaximo
prop_ordenadosPorMaximo :: [[Int]] -> Bool
prop ordenadosPorMaximo xs =
 ordenadosPorMaximo ys == ordenadosPorMaximo2 ys
 where ys = filter (not . null) xs
-- Comprobación
    λ> verificaOrdenadosPorMaximo
     +++ OK, passed 100 tests.
```

La bandera tricolor

Enunciado

El problema de la bandera tricolor consiste en lo siguiente: Dada un lista de objetos xs que pueden ser rojos, amarillos o morados, se pide devolver una lista ys que contiene los elementos de xs, primero los rojos, luego los amarillos y por último los morados.

Se pide definir el tipo de dato Color para representar los colores con los constructores R, A y M correspondientes al rojo, amarillo y morado y la función

```
banderaTricolor :: [Color] -> [Color]
```

tal que (banderaTricolor xs) es la bandera tricolor formada con los elementos de xs. Por ejemplo,

```
\begin{array}{lll} \text{bandera} & [\text{M}, \text{R}, \text{A}, \text{A}, \text{R}, \text{R}, \text{A}, \text{M}, \text{M}] & == & [\text{R}, \text{R}, \text{R}, \text{A}, \text{A}, \text{M}, \text{M}, \text{M}] \\ \text{bandera} & [\text{M}, \text{R}, \text{A}, \text{R}, \text{R}, \text{A}] & == & [\text{R}, \text{R}, \text{R}, \text{A}, \text{A}, \text{M}] \end{array}
```

```
import Data.List (sort)
import Test.QuickCheck

data Color = R | A | M
```

```
deriving (Show, Eq, Ord, Enum)
-- 1ª definición (con sort):
banderaTricolor :: [Color] -> [Color]
banderaTricolor = sort
-- 2ª definición (por comprensión):
banderaTricolor2 :: [Color] -> [Color]
banderaTricolor2 xs =
  [x \mid x \leftarrow xs, x == R] ++ [x \mid x \leftarrow xs, x == A] ++ [x \mid x \leftarrow xs, x == M]
-- 3ª definición (por comprensión y concat):
banderaTricolor3 :: [Color] -> [Color]
banderaTricolor3 xs =
  concat [[x \mid x \leftarrow xs, x == c] \mid c \leftarrow [R,A,M]]
-- 4ª definición (por recursión):
banderaTricolor4 :: [Color] -> [Color]
banderaTricolor4 xs = aux xs ([],[],[])
 where aux [] (rs,as,ms) = rs ++ as ++ ms
        aux (R:ys) (rs,as,ms) = aux ys (R:rs,
                                                  as,
                                                        ms)
        aux (A:ys) (rs,as,ms) = aux ys (rs, A:as,
                                                        ms)
        aux (M:ys) (rs,as,ms) = aux ys (rs,
                                                  as, M:ms)
-- 5ª definición (por recursión):
banderaTricolor5 :: [Color] -> [Color]
banderaTricolor5 xs = aux xs (0,0,0)
 where aux [] (as,rs,ms) = replicate rs R ++
                                 replicate as A ++
                                 replicate ms M
        aux (A:ys) (as,rs,ms) = aux ys (1+as,
                                                  rs,
                                                        ms)
        aux (R:ys) (as,rs,ms) = aux ys (as, 1+rs,
                                                        ms)
        aux (M:ys) (as,rs,ms) = aux ys (as,
                                                  rs, 1+ms
-- Equivalencia
-- =========
instance Arbitrary Color where
  arbitrary = elements [R,A,M]
```

```
prop_banderaTricolor :: [Color] -> Bool
prop banderaTricolor xs =
  all (== banderaTricolor xs)
      [f xs | f <- [ banderaTricolor2</pre>
                   , banderaTricolor3
                    , banderaTricolor4
                    , banderaTricolor5]]
verifica banderaTricolor :: IO ()
verifica banderaTricolor =
  quickCheck prop_banderaTricolor
-- La comprobación es
      λ> verifica banderaTricolor
      +++ OK, passed 100 tests.
-- Comparaciones:
      ghci> bandera n = concat [replicate n c | c <- [M,R,A]]</pre>
      ghci> length (banderaTricolor (bandera 1000000))
      3000000
      (2.65 secs, 312128100 bytes)
      ghci> length (banderaTricolor2 (bandera 1000000))
      3000000
      (7.78 secs, 512387912 bytes)
      ghci> length (banderaTricolor3 (bandera 1000000))
      3000000
      (7.84 secs, 576080444 bytes)
      ghci> length (banderaTricolor4 (bandera 1000000))
      3000000
      (3.76 secs, 476484220 bytes)
      ghci> length (banderaTricolor5 (bandera 1000000))
      3000000
      (4.45 secs, 622205356 bytes)
```

Elementos minimales

Enunciado

Definir la función

```
minimales :: Eq a => [[a]] -> [[a]]
```

tal que (minimales xss) es la lista de los elementos de xss que no están contenidos en otros elementos de xss. Por ejemplo,

```
minimales [[1,3],[2,3,1],[3,2,5]] == [[2,3,1],[3,2,5]] minimales [[1,3],[2,3,1],[3,2,5],[3,1]] == [[2,3,1],[3,2,5]]
```

```
import Data.List (delete, nub)

minimales :: Eq a => [[a]] -> [[a]]

minimales xss =
   [xs | xs <- xss, [ys | ys <- xss, subconjuntoPropio xs ys] == []]

-- (subconjuntoPropio xs ys) se verifica si xs es un subconjunto propio
-- de ys. Por ejemplo,
-- subconjuntoPropio [1,3] [3,1,3] == False
-- subconjuntoPropio [1,3,1] [3,1,2] == True</pre>
```

```
subconjuntoPropio :: Eq a => [a] -> [a] -> Bool
subconjuntoPropio xs ys = subconjuntoPropio' (nub xs) (nub ys)
where
    subconjuntoPropio' _ [] = False
    subconjuntoPropio' [] _ = True
    subconjuntoPropio' (x:xs') ys' =
        x `elem` ys' && subconjuntoPropio xs' (delete x ys')
```

Mastermind

Enunciado

El Mastermind es un juego que consiste en deducir un código numérico formado por una lista de números distintos. Cada vez que se empieza una partida, el programa debe elegir un código, que será lo que el jugador debe adivinar en la menor cantidad de intentos posibles. Cada intento consiste en una propuesta de un código posible que propone el jugador, y una respuesta del programa. Las respuestas le darán pistas al jugador para que pueda deducir el código.

Estas pistas indican cuán cerca estuvo el número propuesto de la solución a través de dos valores: la cantidad de aciertos es la cantidad de dígitos que propuso el jugador que también están en el código en la misma posición. La cantidad de coincidencias es la cantidad de digitos que propuso el jugador que también están en el código pero en una posición distinta.

Por ejemplo, si el código que eligió el programa es el [2,6,0,7], y el jugador propone el [1,4,0,6], el programa le debe responder un acierto (el 0, que está en el código original en el mismo lugar, el tercero), y una coincidencia (el 6, que también está en el código original, pero en la segunda posición, no en el cuarto como fue propuesto). Si el jugador hubiera propuesto el [3,5,9,1], habría obtenido como respuesta ningún acierto y ninguna coincidencia, ya que no hay números en común con el código original, y si se obtienen cuatro aciertos es porque el jugador adivinó el código y ganó el juego.

Definir la función

```
mastermind :: [Int] -> [Int] -> (Int,Int)
```

tal que (mastermind xs ys) es el par formado por los números de aciertos y de coincidencias entre xs e ys. Por ejemplo,

```
mastermind [2,6,0,7] [1,4,0,6] == (1,1)
mastermind [2,6,0,7] [3,5,9,1] == (0,0)
mastermind [2,6,0,7] [1,6,0,4] == (2,0)
mastermind [2,6,0,7] [2,6,0,7] == (4,0)
```

```
import Test.QuickCheck
import Data.List (nub)
-- 1ª solución (por comprensión):
mastermind :: [Int] -> [Int] -> (Int,Int)
mastermind xs ys =
  (length (aciertos xs ys),length (coincidencias xs ys))
-- (aciertos xs ys) es la lista de aciertos entre xs e ys. Por ejemplo,
      aciertos [2,6,0,7] [1,4,0,6] == [0]
aciertos :: Eq a => [a] -> [a] -> [a]
aciertos xs ys = [x \mid (x,y) \leftarrow zip xs ys, x == y]
-- (coincidencia xs ys) es la lista de coincidencias entre xs e ys. Por
-- ejemplo,
      coincidencias [2,6,0,7] [1,4,0,6] == [6]
coincidencias :: Eq a => [a] -> [a] -> [a]
coincidencias xs vs =
  [x | x <- xs, x `elem` ys, x `notElem` zs]</pre>
 where zs = aciertos xs ys
-- 2ª solución (por recursión):
mastermind2 :: [Int] -> [Int] -> (Int,Int)
mastermind2 xs ys = aux xs ys
 where aux [] [] = (0,0)
        aux (x:xs') (z:zs)
            | x == z
                       = (a+1,b)
            \mid x \text{ `elem` ys = (a,b+1)}
            | otherwise = (a,b)
```

```
where (a,b) = aux xs' zs
        aux _ _ = error "Imposible"
-- 3ª solución:
mastermind3 :: [Int] -> [Int] -> (Int,Int)
mastermind3 xs ys = (nAciertos,nCoincidencias)
    where nAciertos = length [(x,y) | (x,y) \leftarrow zip xs ys, x == y]
          nCoincidencias = length (xs++ys) - length (nub (xs++ys)) - nAciertos
-- Equivalencia
-- =========
prop_mastermind :: [Int] -> [Int] -> Bool
prop mastermind xs ys =
  all (== mastermind cs ds)
      [f cs ds | f <- [ mastermind2</pre>
                     , mastermind3]]
 where as = nub xs
        bs = nub ys
        n = min (length as) (length bs)
        cs = take n as
        ds = take n bs
verifica mastermind :: IO ()
verifica_mastermind =
  quickCheck prop_mastermind
-- La comprobación es
     λ> verifica mastermind
     +++ OK, passed 100 tests.
```

Primos consecutivos con media capicúa

Enunciado

La pasada semana, Antonio Roldán publicó en Twitter la siguiente observación:

Los pares de primos consecutivos (97,101) y (109,113) son los más pequeños, con promedio capicúa con más de una cifra: (97+101)/2=99 y (109+113)/2=111.

A partir de ella, se propone el ejercicio de hoy.

Definir la constante

```
primosConsecutivosConMediaCapicua :: [(Int,Int,Int)]
```

tal que primosConsecutivosConMediaCapicua es la lista de las ternas (x,y,z) tales que x e y son primos consecutivos tales que su media, z, es capicúa. Por ejemplo,

```
ghci> take 5 primosConsecutivosConMediaCapicua
[(3,5,4),(5,7,6),(7,11,9),(97,101,99),(109,113,111)]
```

Calcular cuántos hay anteriores a 2014.

Anagramas

Enunciado

Una palabra es una anagrama de otra si se puede obtener permutando sus letras. Por ejemplo, mora y roma son anagramas de amor.

Definir la función

```
anagramas :: String -> [String] -> [String]
```

tal que (anagramas x ys) es la lista de los elementos de ys que son anagramas de x. Por ejemplo,

```
ghci> anagramas "amor" ["Roma","mola","loma","moRa", "rama"]
["Roma","moRa"]
ghci> anagramas "rama" ["aMar","amaRa","roMa","marr","aRma"]
["aMar","aRma"]
```

```
import Data.List (sort)
import Data.Char (toLower)
import Data.Function (on)
-- 1<sup>a</sup> definición (por recursión)
```

```
- - -----
anagramas :: String -> [String] -> [String]
anagramas [] = []
anagramas x (y:ys)
  \mid sonAnagramas x y = y : anagramas x ys
  | otherwise
              = anagramas x ys
-- (sonAnagramas xs ys) se verifica si xs e ys son anagramas. Por
-- ejemplo,
     sonAnagramas "amor" "Roma" == True
     sonAnagramas "amor" "mola" == False
sonAnagramas :: String -> String -> Bool
sonAnagramas xs ys =
  sort (map toLower xs) == sort (map toLower ys)
-- 2ª definición de sonAnagramas
sonAnagramas2 :: String -> String -> Bool
sonAnagramas2 xs ys =
  (sort . map toLower) xs == (sort . map toLower) ys
-- 3ª definición de sonAnagramas (con on)
sonAnagramas3 :: String -> String -> Bool
sonAnagramas3 = (==) `on` (sort . map toLower)
-- Nota. En la definición anterior se usa la función on ya que
-- (f `on` g) x y
-- es equivalente a
-- f(g x)(g y)
-- Por ejemplo,
     ghci> ((*) `on` (+2)) 3 4
     30
-- 2ª definición (por comprensión)
- - -----
anagramas2 :: String -> [String] -> [String]
anagramas2 x ys = [y \mid y \leftarrow ys, sonAnagramas x y]
-- 3º definición (con filter y sin el 2º argumento)
```

Primos equidistantes

Enunciado

Definir la función

```
primosEquidistantes :: Integer -> [(Integer,Integer)]
```

tal que (primosEquidistantes k) es la lista de los pares de primos cuya diferencia es k. Por ejemplo,

```
take 3 (primosEquidistantes 2) == [(3,5),(5,7),(11,13)]
take 3 (primosEquidistantes 4) == [(7,11),(13,17),(19,23)]
take 3 (primosEquidistantes 6) == [(23,29),(31,37),(47,53)]
take 3 (primosEquidistantes 8) == [(89,97),(359,367),(389,397)]
```

```
import Data.Numbers.Primes (primes)

-- 1@ solución
-- =========

primosEquidistantes :: Integer -> [(Integer,Integer)]
primosEquidistantes k = aux primos
  where aux (x:y:ps) | y - x == k = (x,y) : aux (y:ps)
```

```
| otherwise = aux (y:ps)
-- (primo x) se verifica si x es primo. Por ejemplo,
      primo 7 == True
      primo 8 == False
primo :: Integer -> Bool
primo x = [y \mid y \leftarrow [1..x], x \text{ 'rem' } y == 0] == [1,x]
-- primos es la lista de los números primos. Por ejemplo,
      take 10 primos == [2,3,5,7,11,13,17,19,23,29]
primos :: [Integer]
primos = 2 : [x | x \leftarrow [3,5..], primo x]
-- 2ª solución
-- =========
primosEquidistantes2 :: Integer -> [(Integer,Integer)]
primosEquidistantes2 k = aux primes
  where aux (x:y:ps) \mid y - x == k = (x,y) : aux (y:ps)
                      | otherwise = aux (y:ps)
-- 3ª solución
-- =========
primosEquidistantes3 :: Integer -> [(Integer,Integer)]
primosEquidistantes3 k =
  [(x,y) \mid (x,y) \leftarrow zip primes (tail primes)]
         , y - x == k
-- Comparación de eficiencia
      \lambda> primosEquidistantes 2 !! 200
      (9677, 9679)
      (6.30 secs, 896,925,344 bytes)
      λ> primosEquidistantes2 2 !! 200
      (9677, 9679)
      (0.02 secs, 3,622,808 bytes)
      λ> primosEquidistantes3 2 !! 200
      (9677, 9679)
```

```
-- (0.03 secs, 6,398,168 bytes)

-- λ> primosEquidistantes2 2 !! 20000

-- (2840447,2840449)

-- (2.46 secs, 1,142,499,552 bytes)

-- λ> primosEquidistantes3 2 !! 20000

-- (2840447,2840449)

-- (4.32 secs, 2,207,828,128 bytes)
```

Suma si todos los valores son justos

Enunciado

Definir la función

```
sumaSiTodosJustos :: (Num a, Eq a) => [Maybe a] -> Maybe a
```

tal que (sumaSiTodosJustos xs) es justo la suma de todos los elementos de xs si todos son justos (es decir, si Nothing no pertenece a xs) y Nothing en caso contrario. Por ejemplo,

```
sumaSiTodosJustos [Just 2, Just 5] == Just 7
sumaSiTodosJustos [Just 2, Just 5, Nothing] == Nothing
```

```
import Data.Maybe
import Test.QuickCheck
-- 1@ solución
-- ===========
sumaSiTodosJustos :: (Num a, Eq a) => [Maybe a] -> Maybe a
```

```
sumaSiTodosJustos xs
 \mid todosJustos xs = Just (sum [x | (Just x) <- xs])
 | otherwise = Nothing
-- (todosJustos xs) se verifica si todos los elementos de xs son justos
-- (es decir, si Nothing no pertenece a xs) y Nothing en caso
-- contrario. Por ejemplo,
     todosJustos [Just 2, Just 5]
                                         == True
     todosJustos [Just 2, Just 5, Nothing] == False
todosJustos :: Eq a => [Maybe a] -> Bool
todosJustos = notElem Nothing
-- 2ª solución
-- ========
sumaSiTodosJustos2 :: (Num a, Eq a) => [Maybe a] -> Maybe a
sumaSiTodosJustos2 xs
  \mid todosJustos2 xs = Just (sum [x | (Just x) <- xs])
 todosJustos2 :: Eq a => [Maybe a] -> Bool
todosJustos2 = all isJust
-- 3ª solución
-- =========
sumaSiTodosJustos3 :: (Num a, Eq a) => [Maybe a] -> Maybe a
sumaSiTodosJustos3 xs
  | todosJustos xs = Just (sum [fromJust x | x <- xs])</pre>
 -- 4ª solución
-- =========
sumaSiTodosJustos4 :: (Num a, Eq a) => [Maybe a] -> Maybe a
sumaSiTodosJustos4 xs
 | todosJustos xs = Just (sum (catMaybes xs))
 | otherwise = Nothing
-- 5ª solución
```

```
-- ========
sumaSiTodosJustos5 :: (Num a, Eq a) => [Maybe a] -> Maybe a
sumaSiTodosJustos5 xs = suma (sequence xs)
 where suma Nothing = Nothing
        suma (Just ys) = Just (sum ys)
-- Nota. En la solución anterior se usa la función
      sequence :: Monad m \Rightarrow [m \ a] \rightarrow m \ [a]
-- tal que (sequence xs) es la mónada obtenida evaluando cada una de las
-- de xs de izquierda a derecha. Por ejemplo,
      sequence [Just 2, Just 5] == Just [2,5]
      sequence [Just 2, Nothing] == Nothing
     sequence [[2,4],[5,7]]
                             == [[2,5],[2,7],[4,5],[4,7]]
     sequence [[2,4],[5,7],[6]] = [[2,5,6],[2,7,6],[4,5,6],[4,7,6]]
     sequence [[2,4],[5,7],[]] == []
-- 6ª solución
-- =========
sumaSiTodosJustos6 :: (Num a, Eq a) => [Maybe a] -> Maybe a
sumaSiTodosJustos6 xs = fmap sum (sequence xs)
-- 7ª solución
-- =========
sumaSiTodosJustos7 :: (Num a, Eq a) => [Maybe a] -> Maybe a
sumaSiTodosJustos7 = fmap sum . sequence
-- Equivalencia
-- ========
-- La propiedad es
prop_sumaSiTodosJustos :: [Maybe Int] -> Bool
prop sumaSiTodosJustos xs =
  all (== sumaSiTodosJustos xs)
      [f xs | f <- [ sumaSiTodosJustos2</pre>
                   , sumaSiTodosJustos3
                   , sumaSiTodosJustos4
                   , sumaSiTodosJustos5
```

```
, sumaSiTodosJustos6
, sumaSiTodosJustos7
]]
```

```
    La comprobación es
    λ> quickCheck prop_sumaSiTodosJustos
    +++ 0K, passed 100 tests.
```

Matrices de Toeplitz

Enunciado

Una matriz de Toeplitz es una matriz cuadrada que es constante a lo largo de las diagonales paralelas a la diagonal principal. Por ejemplo,

la primera es una matriz de Toeplitz y la segunda no lo es.

Las anteriores matrices se pueden definir por

```
ejl, ej2 :: Array (Int,Int) Int
ejl = listArray ((1,1),(4,4)) [2,5,1,6,4,2,5,1,7,4,2,5,9,7,4,2]
ej2 = listArray ((1,1),(4,4)) [2,5,1,6,4,2,6,1,7,4,2,5,9,7,4,2]
```

Definir la función

```
esToeplitz :: Eq a => Array (Int,Int) a -> Bool
```

tal que (esToeplitz p) se verifica si la matriz p es de Toeplitz. Por ejemplo,

```
esToeplitz ej1 == True
esToeplitz ej2 == False
```

Máximos locales

Enunciado

Un máximo local de una lista es un elemento de la lista que es mayor que su predecesor y que su sucesor en la lista. Por ejemplo, 5 es un máximo local de [3,2,5,3,7,7,1,6,2] ya que es mayor que 2 (su predecesor) y que 3 (su sucesor).

Definir la función

```
maximosLocales :: Ord a => [a] -> [a]
```

tal que (maximosLocales xs) es la lista de los máximos locales de la lista xs. Por ejemplo,

```
\begin{array}{lll} \text{maximosLocales} & [3,2,5,3,7,7,1,6,2] & == & [5,6] \\ \text{maximosLocales} & [1..100] & == & [] \\ \text{maximosLocales} & \text{"adbpmqexyz"} & == & \text{"dpq"} \end{array}
```

```
import Test.QuickCheck
```

```
-- 1º solución
maximosLocales :: Ord a => [a] -> [a]
```

```
maximosLocales (x:y:z:xs) | y > x \& y > z = y: maximosLocales (z:xs)
                           | otherwise = maximosLocales (y:z:xs)
maximosLocales
                                            = []
-- 2ª solución
maximosLocales2 :: Ord a => [a] -> [a]
maximosLocales2 xs =
  [y \mid (x,y,z) \leftarrow zip3 \ xs \ (tail \ xs) \ (drop 2 \ xs), \ y > x, \ y > z]
-- Equivalencia
-- =========
-- La propiedad es
prop maximosLocales :: [Int] -> Bool
prop maximosLocales xs =
  maximosLocales xs == maximosLocales2 xs
-- La comprobación es
      λ> quickCheck prop maximosLocales
      +++ 0K, passed 100 tests.
-- Comparación de eficiencia
 - -----
      \lambda> maximosLocales [1..10^6]
      Γ1
      (2.50 secs, 464,396,256 bytes)
      \lambda> maximosLocales2 [1..10^6]
      []
      (1.36 secs, 280,395,256 bytes)
      \lambda> maximosLocales (concat (replicate 10 [1..10^5]))
      [100000, 100000, 100000, 100000, 100000, 100000, 100000, 100000]
      (2.55 secs, 455,647,560 bytes)
      λ> maximosLocales2 (concat (replicate 10 [1..10<sup>5</sup>]))
      [100000, 100000, 100000, 100000, 100000, 100000, 100000, 100000]
      (1.34 secs, 271,650,096 bytes)
```

Lista cuadrada

Enunciado

Definir la función

```
listaCuadrada :: Int -> a -> [a] -> [[a]]
```

tal que (listaCuadrada n x xs) es una lista de n listas de longitud n formadas con los elementos de xs completada con x, si no xs no tiene suficientes elementos. Por ejemplo,

```
listaCuadrada 3 7 [0,3,5,2,4] == [[0,3,5],[2,4,7],[7,7,7]]
listaCuadrada 3 7 [0..] == [[0,1,2],[3,4,5],[6,7,8]]
listaCuadrada 2 'p' "eva" == ["ev","ap"]
listaCuadrada 2 'p' ['a'..] == ["ab","cd"]
```

```
import Test.QuickCheck
```

```
take n (grupos n (xs ++ repeat x))
-- (grupos n xs) es la lista obtenida agrupando los elementos de xs en
-- grupos de n elementos, salvo el último que puede tener menos. Por
-- ejemplo,
     grupos \ 2 \ [4,2,5,7,6] == [[4,2],[5,7],[6]]
     take 3 (grupos 3 [1..]) == [[1,2,3],[4,5,6],[7,8,9]]
grupos :: Int -> [a] -> [[a]]
grupos [] = []
grupos n xs = take n xs : grupos n (drop n xs)
-- 2ª solución
-- ========
listaCuadrada2 :: Int -> a -> [a] -> [[a]]
listaCuadrada2 n x xs =
 take n [take n (drop m xs ++ repeat x)
         | m < - [0, n..n*n]]
-- 3ª solución
-- ========
listaCuadrada3 :: Int -> a -> [a] -> [[a]]
listaCuadrada3 n x xs =
 take n [take n ys | ys <- iterate (drop n) (xs ++ repeat x)]
-- 4ª solución
-- =========
listaCuadrada4 :: Int -> a -> [a] -> [[a]]
listaCuadrada4 n x =
 take n . map (take n) . iterate (drop n) . (++ repeat x)
-- Equivalencia
-- =========
-- La propiedad es
prop listaCuadrada :: NonNegative Int -> Int -> [Int] -> Bool
prop_listaCuadrada (NonNegative n) x xs =
 all (== listaCuadrada n x xs)
```

Segmentos de elementos consecutivos

Enunciado

Definir la función

```
segmentos :: (Enum a, Eq a) => [a] -> [[a]]
```

tal que (segmentos xss) es la lista de los segmentos de xss formados por elementos consecutivos. Por ejemplo,

```
segmentos [1,2,5,6,4] == [[1,2],[5,6],[4]]
segmentos [1,2,3,4,7,8,9] == [[1,2,3,4],[7,8,9]]
segmentos "abbccddeeebc" == ["ab","bc","cd","de","e","e","bc"]
```

Nota: Se puede usar la función succ tal que (succ x) es el sucesor de x. Por ejemplo,

```
succ 3 == 4
succ 'c' == 'd'
```

Soluciones

import Test.QuickCheck

```
-- 1ª solución
-- =========
segmentos :: (Enum a, Eq a) => [a] -> [[a]]
segmentos[] = []
segmentos [x] = [[x]]
segmentos (x:xs) \mid y == succ x = (x:y:ys):zs
                | otherwise = [x]:(y:ys):zs
    where ((y:ys):zs) = segmentos xs
-- 2ª solución
-- =========
segmentos2 :: (Enum a, Eq a) => [a] -> [[a]]
segmentos2 [] = []
segmentos2 xs = ys : segmentos2 zs
 where ys = inicial xs
        n = length ys
        zs = drop n xs
-- (inicial xs) es el segmento inicial de xs formado por elementos
-- consecutivos. Por ejemplo,
     inicial [1,2,5,6,4] == [1,2]
     inicial "abccddeeebc" == "abc"
inicial :: (Enum a, Eq a) => [a] -> [a]
inicial [] = []
inicial (x:xs) =
  [y \mid (y,_) \leftarrow takeWhile ((u,v) \rightarrow u == v) (zip (x:xs) [x..])]
```

Valor de un polinomio mediante vectores

Enunciado

Los polinomios se pueden representar mediante vectores usando la librería Data.Array. En primer lugar, se define el tipo de los polinomios (con coeficientes de tipo a) mediante

```
type Polinomio a = Array Int a
```

Como ejemplos, definimos el polinomio

```
ej_pol1 :: Array Int Int
ej_pol1 = array (0,4) [(1,2),(2,-5),(4,7),(0,6),(3,0)]
```

que representa a $2x - 5x^2 + 7x^4 + 6$ y el polinomio

```
ej_pol2 :: Array Int Double
ej_pol2 = array (0,4) [(1,2),(2,-5.2),(4,7),(0,6.5),(3,0)]
```

que representa a $2x - 5.2x^2 + 7x^4 + 6.5$

Definir la función

```
valor :: Num a => Polinomio a -> a -> a
```

tal que (valor p b) es el valor del polinomio p en el punto b. Por ejemplo,

```
valor ej_pol1 0 == 6
valor ej_pol1 1 == 10
valor ej_pol1 2 == 102
valor ej_pol2 0 == 6.5
valor ej_pol2 1 == 10.3
valor ej_pol2 3 == 532.7
```

```
import Data.Array

type Polinomio a = Array Int a

ej_pol1, ej_pol2 :: Array Int Double
ej_pol1 = array (0,4) [(1,2),(2,-5),(4,7),(0,6),(3,0)]
ej_pol2 = array (0,4) [(1,2),(2,-5.2),(4,7),(0,6.5),(3,0)]

-- 1ª solución

valor :: Num a => Polinomio a -> a -> a

valor p b = sum [(p!i)*b^i | i <- [0..n]]
   where (_,n) = bounds p

-- 2ª solución

valor2 :: Num a => Polinomio a -> a -> a

valor2 p b = sum [v*b^i | (i,v) <- assocs p]</pre>
```

Ramas de un árbol

Enunciado

Los árboles se pueden representar mediante el siguiente tipo de datos

Por ejemplo, los árboles

```
1 3

/\ /\ /|\

2 3 /|\

| 5 4 7

4 | /\

6 2 1
```

se representan por

```
ej1, ej2 :: Arbol Int
ej1 = N 1 [N 2 [],N 3 [N 4 []]]
ej2 = N 3 [N 5 [N 6 []], N 4 [], N 7 [N 2 [], N 1 []]
```

Definir la función

```
ramas :: Arbol b -> [[b]]
```

tal que (ramas a) es la lista de las ramas del árbol a. Por ejemplo,

```
ramas ej1 == [[1,2],[1,3,4]]
ramas ej2 == [[3,5,6],[3,4],[3,7,2],[3,7,1]]
```

```
import Test.QuickCheck
data Arbol a = N a [Arbol a]
                deriving Show
ej1, ej2 :: Arbol Int
ej1 = N 1 [N 2 [], N 3 [N 4 []]]
ej2 = N 3 [N 5 [N 6 []], N 4 [], N 7 [N 2 [], N 1 []]]
-- 1ª solución
ramas :: Arbol b -> [[b]]
ramas (N \times []) = [[x]]
ramas (N \times as) = [X \times xs \mid a \leftarrow as, xs \leftarrow ramas a]
-- 2ª solución
ramas2 :: Arbol b -> [[b]]
ramas2 (N x []) = [[x]]
ramas2 (N x as) = concat (map (map (x:)) (map ramas2 as))
-- 3ª solución
ramas3 :: Arbol b -> [[b]]
ramas3 (N x []) = [[x]]
ramas3 (N x as) = concatMap (map (x:)) (map ramas3 as)
-- 4ª solución
ramas4 :: Arbol b -> [[b]]
ramas4 (N x []) = [[x]]
ramas4 (N x as) = concatMap (map (x :) . ramas4) as
-- 5ª solución
ramas5 :: Arbol a -> [[a]]
ramas5 (N x []) = [[x]]
```

```
ramas5 (N x xs) = map ramas5 xs >>= map (x:)
-- Equivalencia
-- =========
-- Generador de árboles.
      > sample ((gen Arbol 5) :: Gen (Arbol Int))
      N 0 [N 0 []]
      N(-2)[]
     N 4 []
     N 2 [N 4 []]
     N 8 []
     N (-2) [N (-9) [],N 7 []]
     N 11 []
     N (-11) [N 4 [],N 14 []]
     N 10 [N (-3) [],N 13 []]
     N 12 [N 11 []]
      N 20 [N (-18) [],N (-13) []]
gen Arbol :: Arbitrary a => Int -> Gen (Arbol a)
gen Arbol m = do
 x <- arbitrary
  n \leftarrow choose (0, m \dot v) 2)
  xs <- vectorOf n (gen_Arbol (m `div` 4))</pre>
  return (N x xs)
-- Incluye los árboles en Arbitrary.
instance Arbitrary a => Arbitrary (Arbol a) where
  arbitrary =
    sized gen_Arbol
-- La propiedad es
prop ramas :: Arbol Int -> Bool
prop_ramas a =
  all (== ramas a)
      [f a | f <- [ ramas2</pre>
                   , ramas3
                   , ramas4
                   , ramas5
                  ]]
```

```
-- La comprobación es-- λ> quickCheck prop_ramas-- +++ OK, passed 100 tests.
```

Alfabeto comenzado en un carácter

Enunciado

Definir la función

```
alfabetoDesde :: Char -> String
```

tal que (alfabetoDesde c) es el alfabeto, en minúscula, comenzando en el carácter c, si c es una letra minúscula y comenzando en 'a', en caso contrario. Por ejemplo,

```
alfabetoDesde 'e' == "efghijklmnopqrstuvwxyzabcd"
alfabetoDesde 'a' == "abcdefghijklmnopqrstuvwxyz"
alfabetoDesde '7' == "abcdefghijklmnopqrstuvwxyz"
alfabetoDesde '{' == "abcdefghijklmnopqrstuvwxyz"
alfabetoDesde 'B' == "abcdefghijklmnopqrstuvwxyz"
```

```
import Data.Char (isAsciiLower, ord)
import Data.Tuple (swap)
import Test.QuickCheck
```

```
-- 1ª solución
alfabetoDesde :: Char -> String
alfabetoDesde c =
  dropWhile (<c) ['a'..'z'] ++ takeWhile (<c) ['a'..'z']</pre>
-- 2ª solución
alfabetoDesde2 :: Char -> String
alfabetoDesde2 c = ys ++ xs
  where (xs,ys) = span (<c) ['a'...'z']
-- 3ª solución
alfabetoDesde3 :: Char -> String
alfabetoDesde3 c = ys ++ xs
 where (xs,ys) = break (==c) ['a'...'z']
-- 4ª solución
alfabetoDesde4 :: Char -> String
alfabetoDesde4 = uncurry (++) . swap . flip span ['a'..'z'] . (>)
-- Ejemplo de cálculo:
      alfabetoDesde4 'e'
     = (uncurry (++) . swap . flip span ['a'..'z'] . (>)) 'e'
     = (uncurry (++) . swap) ("abcd", "efghijklmnopqrstuvwxyz")
     = uncurry (++) ("efghijklmnopqrstuvwxyz", "abcd")
     = (++) "efghijklmnopgrstuvwxyz" "abcd"
     = "efghijklmnopgrstuvwxyzabcd"
-- 5ª solución
alfabetoDesde5 :: Char -> String
alfabetoDesde5 = uncurry (flip (++)) . (`break` ['a'..'z']) . (==)
-- Ejemplo de cálculo:
      alfabetoDesde5 'e'
     = (uncurry (flip (++)) . (`break` ['a'..'z']) . (==)) 'e'
     = uncurry (flip (++)) ("abcd", "efghijklmnopgrstuvwxyz")
     = flip (++) "abcd" "efghijklmnopqrstuvwxyz"
     = "efghijklmnopqrstuvwxyz" ++ "abcd"
     = "efghijklmnopgrstuvwxyzabcd"
-- 6ª solución
```

```
alfabetoDesde6 :: Char -> String
alfabetoDesde6 c
 | c >= 'a' \&\& c <= 'z' = [c..'z'] ++ ['a'..pred c]
  | otherwise = ['a'...'z']
-- 7ª solución
alfabetoDesde7 :: Char -> String
alfabetoDesde7 c
  | isAsciiLower c = [c..'z'] ++ ['a'..pred c]
  | otherwise = ['a'...'z']
-- Equivalencia
-- =========
-- La propiedad es
prop alfabetoDesde :: Char -> Bool
prop_alfabetoDesde c =
 all (== alfabetoDesde c)
      [f c | f <- [ alfabetoDesde2</pre>
                  , alfabetoDesde3
                  , alfabetoDesde4
                  , alfabetoDesde5
                  , alfabetoDesde6
                  , alfabetoDesde7
                  ]]
-- La comprobación es
     λ> quickCheck prop_alfabetoDesde
     +++ OK, passed 100 tests.
```